U. S. DEPARTMENT OF COMMERCE
NATIONAL OCEANIC AND ATMOSPHERIC ADMINISTRATION
NATIONAL WEATHER SERVICE

NOAA Technical Memorandum NWSTM PR-19

A PROGRAM TO COMPUTE TURBULENCE IN THE VICINITY OF LEE WAVES DOWNSTREAM OF SELECTED MOUNTAINS IN THE HAWAIIAN ISLANDS

Lawrence D. Burroughs*

Techniques Development Laboratory
National Weather Service, NOAA
Silver Spring, MD 20910

October 1978

PROPERTY OF
NOAA Library E/OC43
7600 Sand Point Way NE
Seattle WA 98115-0070

*Formerly with WSFO Honolulu, Hawaii

CORRECTION SHEET TO NOAA TECHNICAL MEMO

"A Program to Compute Turbulence in the Vicinity of Lee Waves Downstream of Selected Mountains in the Hawaiian Islands"

1. In equations (7), (8), and (9): change h to a.

2. Page 2, para 2, 4th line: Change equation to: $\ell_1^2 - \ell_2^2 > \pi^2/4h^2$.

3. Page 3, para 2, 2nd line: Change equation to: $\pi^2/4h^2$, ...

4. Page 3, para 2, 8th line: Change sentence to read: ....flow, a is height of mountain above the terrain, b is the half width of the mountain.....

5. Page 3 para 2, 10th line: Change sentence to read: ....at distance x from the crest of the mountain ridge and h is the mountain height above sea level, U_1 is the wind....

6. Page 5, para 2, 11th line: Insert the following sentence ahead of "The following table applies".

   a is determined by substracting ρ from h.

7. Page 5, Table 1: Change column h to a. Change values under column a to:

Koolaus/NE	692 m
Koolaus/SW	692 m
Mt. Kaala/SW	830 m
Mt. Haleakala/W	2559 m

8. Page 12: Change line 1190 LET H=814 to 1190 LET H=692
   1220 LET H=814 to 1120 LET H=692
   1250 LET H=1104 to 1250 LET H=830
   1280 LET H=2803 to 1280 LET H=2559

TABLE OF CONTENTS

		Page
I.	INTRODUCTION AND HISTORY	1
II.	THEORY	1
III.	APPLICATION	4
IV.	RESULTS AND PROBLEMS	6
V.	NOTABLE MISCELLANY	6
VI.	SUGGESTED PROJECTS	8
VII.	REFERENCES	9
	Appendix I - Program Listing	11
	Appendix II - Sample Output	14

ABSTRACT

With the advent of the geosynchronous satellite mountain wave clouds have become visible downstream of several islands in the Hawaiian Island chain. Turbulence associated with these waves poses a threat to aviation safety. This paper reviews the theory of lee wave formation and several applications of that theory with a view to now-casting turbulence categories associated with the waves in the Hawaiian chain. A computer program for minicomputers is described and suggestions for further study are made.

I. INTRODUCTION AND HISTORY

Since data from the SMS-2 satellite became available, mountain wave clouds have been visible downstream of the Koolaus on Oahu and occasionally down-stream of mountain ranges on Kauai, Molokai and Maui. Burroughs and Larson (1978) have documented a couple of well-defined cases of lee waves downstream of the Koolaus on Oahu. Subsequently a paper by Reiter and Foltz (1967) came to light which related the upward vertical motion found in lee waves to documented turbulence reported by specially instrumented planes. Using their formulations it is possible to get a quantitative feel for the actual turbulence in lee waves. Corby and Wallington's (1956) application of Scorer's theories (1949), (1953), (1954) was also studied, and this resulted in the development of a computer program called TURBULENCE. It is designed to compute categories of turbulence in the vicinity of lee waves forming downstream of the Koolaus on the island of Oahu when winds are from the NE or SW quadrants, of Mt. Kaala on the island of Oahu when winds are from the SW quadrant, and of Mt. Haleakala on the island of Maui when winds are from the W quadrant. A program listing and a sample for NE winds are included as Appendices.

II. THEORY

Scorer (1949) develops his theory utilizing dry, invicid, and isentropic stream flow in which there is a small proportionate disturbance of the wind velocity. He uses a two-layer model in which ψ is a stream function with a small disturbance caused by a mountain. The equation of motion is given by

$$\partial^2\psi/\partial Z^2 - (gc^{-2}+Sg^{-1})\psi + (SU^{-2}-(\partial^2U/\partial Z^2)U^{-1}-k^2)\psi = 0 \qquad (1)$$

where $\ell^2 = SU^{-2} - (\partial^2U/\partial Z^2)U^{-1}$ is the Scorer parameter, (2)

$S = g\Theta^{-1}(\partial\theta/\partial Z)$ is the static stability, and U is the wind speed normal to the mountain range. k is the wave number in 2π unit distances in a direction perpendicular to the axis of the mountain range.

Fig. 1 shows the diagram of Scorer's two-layer model to solve the lee wave problem. The horizontal direction is taken normal to the axis of the mountain range positive downstream of the range.



Figure 1

For waves to form $\ell_1^2 - \ell_2^2 > \pi/4h^2$. Since the second term on the right side of (2) above is not important above the friction layer (900 m), it can normally be deleted; therefore,

$$\ell^2 = SU^{-2} \tag{3}$$

and the wavelength is given by

$$\lambda = 2\pi\ell^{-1} = 2\pi US^{-1/2}. \tag{4}$$

Corby and Wallington (1956) have shown that a spectrum of wavelengths occurs in mountain wave events, but the largest wavelength corresponding to the smallest wave number, k, dominates the flow and lies between ℓ_1 and ℓ_2; therefore, they theorize

$$k^2 = \ell_2^2 + L^2\cos\alpha \tag{5}$$

where $L^2 = \ell_1^2 - \ell_2^2$ and k is the actual wave number measured from satellite imagery, for example. The boundary height is given by

$$h = (n\pi - \alpha)/L\sin\alpha \quad n=1,\ 2,\ 3,\ \ldots. \tag{6}$$

Generally n is taken to be 1 since all parameters describing the lee waves are maximized for n = 1.

According to Corby and Wallington (1956), the most favorable conditions for wave formation are when L^2 is slightly greater than $\pi/4h^2$, when L is large and when ℓ_2^2 is small. However, Scorer (1954) points out that even if these criteria are not met or ℓ_2 is not less than ℓ_1, waves may form if an inversion is present. Corby and Wallington's (1956) solution to (1) is given by

$$\rho_z = [-2\pi h b/\exp(kb)]\ (U_1/U_z)\psi_{z,k}(\partial\psi_{1,k}/\partial k)\sin kx \tag{7}$$

where ρ_z is the deviation of a streamline at level z from undisturbed flow, h is height of the mountain, b is the half width of the mountain determined from $\rho = hb^2/(b^2 + x^2)$ where ρ is the height of the terrain at distance x from the crest of the mountain ridge, U_1 is the wind component normal to the mountain crest at gradient level, U_z is the wind component normal to the mountain crest at level z, $\psi_{z,k}$ is the stream function at level z for wave number k, and $\psi_{1,k}$ is the stream function at gradient level for wave number k.

The upward maximum vertical motion is given by

$$w = U_z(\partial \rho_z/\partial x)_{max} = [-2\omega b/\exp(kb)]\ U_1 a_n k \tag{8}$$

where

$$a_n = \psi_{z,k}(\partial \psi_{1,k}/\partial k)^{-1} = \text{amplitude factor}$$

After some re-arranging and further work, Reiter and Foltz (1967) derived the following formulation of (7) for the vertical motion in mountain waves.

$$w_o = -(2\pi\omega b/\exp(kb))U_1 L^2(\sin^2\alpha/(n\pi-\alpha+\tan\alpha)). \tag{9}$$

III. APPLICATION

Alaka (1960) showed

$$\lambda = 2\pi\overline{U}/\overline{S}^{\frac{1}{2}} \tag{10}$$

where the bars indicate averages over large depths in the atmosphere. Since $\overline{S}$ is difficult to calculate, but $\overline{U}$ is relatively easy to calculate, $\overline{S}$ is derived using the following

$$\overline{S} = 4\pi^2\overline{U}^2\lambda^{-2} \tag{11}$$

where $\overline{U}$ is the average wind component normal to the mountain range from the pibal level nearest the crest to the pibal level where the wind maximum occurs, λ is the wavelength measured from the satellite imagery.

Next ℓ_1^2 and ℓ_2^2 are calculated using the wind at the crest and the maximum wind. These are used to determine L^2, then α is determined by re-arranging equation (5) such that

$$\alpha = \arccos(c) \text{ where } c = (k^2-\ell_2^2)L^{-2} \tag{12}$$

but since the mini-computer available doesn't have an inverse cosine function, an identity utilizing the inverse tangent function is used instead. Therefore,

$$\alpha = \arctan\,[(c^{-1})(1-c^2)^{\frac{1}{2}}] \tag{13}$$

The wave number, k, is determined using

$$k = 2\pi\lambda^{-1} \tag{14}$$

where λ is the measured wavelength.

Four sets of mountain parameters have been determined using $\rho = hb^2(b^2 + x^2)^{-1}$: a set for the Koolaus for trade conditions which applies only to the portion of the range not windward of the Waianaes, a set for the Koolaus for SW winds which applies to the portion of the range not leeward of the Waianaes, a set for Mt. Kaala for SW winds, and a set for Mt. Haleakala for W winds. The distance x was chosen to be the distance at which the slope became appreciably more gentle than that for the mountain range itself. Only in the case of a SW wind over the Koolaus would the results be materially changed by changing x. The effect would be to reduce the turbulence forecast and underforecast it. (a is determined by subtracting ρ from h) The following table applies.

RANGE/WIND	b	~~h~~ a	ρ	x
Koolaus/NE	3379m	~~814m~~ 692	400ft	5mi
Koolaus/SW	756m	~~814m~~ 692	400ft	1800m
Mt. Kaala/SW	3702m	~~1104m~~ 830	900ft	4mi
Mt. Haleakala/W	4576m	~~2803m~~ 2559	800ft	8nm

Table 1. Parameters defining the equivalent mountain used in determining upward vertical motion in the lee waves.

Reiter and Foltz (1967) developed a nomograph which gives estimated Clear Air Turbulence intensity levels. In order to develop equations for the curves in the nomograph curve fitting was done using cubic equations. The fitted equations are given below.

$$W_1 = 4.7451\times10^{-4}\lambda^3 - 2.5353\times10^{-2}\lambda^2 + 0.448\lambda + 0.576 \quad (15)$$

$$W_2 = 1.3333\times10^{-4}\lambda^3 - 1.0\times10^{-2}\lambda^2 + 0.386\lambda + 1.0, \quad (16)$$

$$W_3 = 2.5297\times10^{-3}\lambda^3 - 9.1369\times10^{-2}\lambda^2 + 1.24\lambda + 1.0 \quad (17)$$

where $W_{1,2,3}$ are in $msec^{-1}$ and λ is in km. W_1 defines a line separating categories of light and light to moderate turbulence; W_2 defines a line separating categories of light to moderate and moderate to severe turbulence; W_3 defines a line separating categories of moderate to severe and severe turbulence. W_o is then compared to $W_{1,2,3}$ and the resulting category of turbulence is outputted to the user.

IV. RESULTS AND PROBLEMS

In general, the results seem reasonable; however, in the trade wind situation the maximum wind generally occurs below the trade inversion, and the values outputted appear low. Calculated values may indicate light turbulence when, in fact, it's reported to be moderate. This may be due to a variety of reasons including errors in the data or data not being representative--How good are Wheeler soundings? Do Lihue soundings represent the air mass over Oahu? Are the Honolulu pibals representative?--the way S, ℓ_1^2, ℓ_2^2 are calculated and the neglect of $(\partial^2U/\partial Z^2)U^{-1}$ in (2). Errors due to these factors appear to account for about 20 percent.

V. NOTABLE MISCELLANY

During trade wind situations two wind maxima are often apparent: one associated with the northeasterly winds in the lower half of the troposphere, and the other associated with the westerly winds in the upper half of the troposphere with a minimum and wind reversal between. Under these circumstances the number of wind velocities inputted include only those from the crest of the ridge to the first maxima.

In the few cases examined it has been noted that the criteria $L^2 > \pi/4h^2$ has not been met whenever a strong inversion exists, even though waves have been evident. Scorer (1954) notes an inversion alters the criteria for wave formation. To overcome this problem somewhat and to account for the errors previously discussed, the criteria has been altered to $L^2 > 0.9\pi/4h^2$, and a question has been added to help the forecaster decide whether or not to go on. Scorer (1953), (1954) has pointed out several other interesting facts which help to explain what is often seen in the SMS-2 imagery at Honolulu. There is a diurnal variation in wavelength which occurs and may explain why (especially during trade conditions) the waves disappear in the afternoon. ℓ^2 is reduced after sunrise and increased after sunset due to heating and decreasing stability. As the day progresses the wavelength increases until it essentially becomes infinite, i.e., there is no wave, and the criteria $L^2 > \pi/4h^2$ is no longer met. They tend to reappear at dusk. Waves occurring downstream of a single peak have shorter wavelengths and less turbulence than those downstream of an "infinite ridge". The persistence downstream of waves is greater the longer the ridge. The longer the wavelength is; the greater the amplitude is. Also narrower, steeper obstructions induce greater amplitudes (that's why the SW flow over the Koolaus on the island of Oahu can be so dangerous; the amplitude of the waves created is greater; and therefore, the turbulence is greater). Where a temperature inversion exists, the maximum amplitude of the waves will occur below the inversion. Where the amplitude is greatest is where the shear is greatest; and, hence, the turbulence is greatest (Pao and Goldburg, 1969).

VI. SUGGESTED PROJECTS

Several projects which would improve and expand the program's capabilities follow. Lee waves tend to damp out downstream, and a regression equation relating the number of waves downstream to the average wind speed normal to the ridge from the crest of the ridge to the wind maximum should be derived. From this regression equation a damping function can be developed to show the forecaster how far downstream moderate of greater turbulence can be expected. There are days when conditions favorable to formation of lee waves exist, but wave clouds do not form due to lack of moisture or cannot be seen due to mid and high cloud interference. Here again a regression equation should be developed equating the wavelength to the average wind speed (defined above). In both derivations suggested, separate formulae should be developed for strong inversion situations where the speed maximum is below the inversion and for all other cases combined. It may be necessary to further stratify the data by wind direction. Once such regression equations exist, a shift from nowcasting to forecasting can be made because it will no longer be necessary to see wave clouds to do the calculations. The forecasts could then be refined by use of the satellite imagery. The program's utility may also be increased by adding mountain parameters for Kauai, Molokai and the combined effect of the Koolaus and Waianaes. Some means of making the Koolaus and Waianaes into an equivalent range needs to be studied before the latter can be done.

VII. ACKNOWLEDGMENTS

Special thanks are due Ms. Barbara Howerton for typing the manuscript and Prof. Wan-Cheng Chiu for reviewing it. His comments have been very helpful.

VIII. REFERENCES

Alaka, M. A. (Ed.), 1960: The airflow over mountains, WMO Technical Note No. 34. (WMO No. 98, TP. 43), Geneva. World Meteorological Organization, 135 pp.

Burroughs, L. D., and R. N. Larson, 1978: Wave clouds in the vicinity of Oahu Island, Hawaii (in press).

Corby, G. A., and C. E. Wallington, 1956: Airflow over mountains: the lee wave amplitude. Quart. J. Roy. Meteor. Soc., 82, 266-274 pp.

Pao, Y. H. and A. Goldburg, (Eds.), 1969: C.A.T. and its detection. New York, Plenum Press, 542 pp.

Reiter, E. R. and H. P. Foltz, 1967: The prediction of clear air turbulence over mountainous terrain. J. Appl. Meteor., 6, 549-566.

Scorer, R. S., 1949: Theory of waves in the lee of mountains. Quart. J. Roy. Meteor. Soc., 75, 41-56 pp.

______, 1953: Theory of airflow over mountains: II - The flow over a ridge. Quart. J. Roy. Meteor. Soc., 79, 70-83 pp.

______, 1954: Theory of airflow over mountains: III - Airstream characteristics. Quart. J. Roy. Meteor. Soc., 80, 417-428 pp.

PROGRAM LISTING

```
BASIC TURBULENCE
LIST
100 DIM X[25],T[25],S[25],A1[25]
110 PRINT
120 PRINT "THIS PROGRAM GIVES CATAGORIES OF TURBULENCE WHICH MAY"
130 PRINT "BE EXPECTED IN THE VICINITY OF WAVES IN THE LEE OF"
140 PRINT "MOUNTAIN RANGES OR INDIVIDUAL MOUNTAIN PEAKS. IT IS"
150 PRINT "CURRENTLY VALID FOR THE KOOLAUS WHEN WINDS ARE FROM"
160 PRINT "THE NE OR SW QUADRANTS, MT KAALA WHEN WINDS ARE FROM"
170 PRINT "SW QUADRANT AND MT HALEAKALA WHEN WINDS ARE FROM THE"
180 PRINT "W QUADRANT. THE INPUTS TO THE PROGRAM ARE THE NUMBER"
190 PRINT "OF PIBAL WINDS BETWEEN THE CREST OF THE MOUNTAIN"
200 PRINT "AND THE MAX WIND ABOVE THE CREST INCLUSIVE, THE WIND"
210 PRINT "DIRECTION AND SPEED AT EACH LEVEL, AND THE WAVELENGTH"
220 PRINT "MEASURED AS THE NUMBER OF WAVES PER DISTANCE, IE USING"
230 PRINT "AT LEAST 5 WAVE CLOUDS MEASURE THE DISTANCE FROM FIRST"
240 PRINT "TO LAST WAVE AND INPUT THE NUMBER OF WAVES AND THE"
250 PRINT "DISTANCE MEASURED IN NAUTICAL MILES (NM). THE 3000FT"
260 PRINT "WIND IS THE STARTING POINT FOR THE MOUNTAINS ON OAHU."
270 PRINT "THE 10000FT WIND OR WIND NEAREST TO THAT LEVEL ABOVE"
280 PRINT "THE SUMMIT OF HALEAKALA IS THE STARTING POINT ON MAUI."
290 PRINT "THE LAST WIND INPUTTED SHOULD ALWAYS BE THE MAX WIND."
300 PRINT "USE THE BEST WINDS AVAILABLE. YOU MAY USE UP TO 25"
310 PRINT "WIND VELOCITIES. USE KNOTS FOR ALL WIND SPEEDS."
320 PRINT "WHAT IS THE NUMBER OF WIND VELOCITIES YOU WISH TO USE?"
330 PRINT
340 INPUT M
350 PRINT
360 PRINT "GIVE ME YOUR VELOCITIES AS DIRECTION, SPEED RETURN."
370 PRINT
380 FOR I=1 TO M
390   INPUT X[I],T[I]
400   LET S[I]=T[I]*6080/(3600*3.2808)
410 NEXT I
420 PRINT
430 PRINT "GIVE ME NUMBER OF WAVES, DISTANCE MEASURED RETURN"
440 PRINT
450 INPUT N,D
460 PRINT
470 LET W=D*6080/(N*3.2808)
480 LET W=W/1000
490 LET W1=4.7451E-04*W*W*W-2.5353E-02*W*W+.448*W+.576
500 LET W2=1.3333E-04*W*W*W-1E-02*W*W+.386*W+1
510 LET W3=2.5297E-03*W*W*W-9.1369E-02*W*W+1.24*W+1
520 LET W=W*1000
530 PRINT
540 PRINT "GIVE ME WIND DIRECTION. IF NE QUAD TYPE 1, IF SW QUAD"
550 PRINT "TYPE 2, IF WAVES APPEAR TO BE FORMING IN THE LEE OF MT"
560 PRINT "KAALA ONLY TYPE 3, IF W QUAD TYPE 4."
570 PRINT
580 INPUT Z
590 IF Z=1 GOTO 630
600 IF Z=2 GOTO 730
610 IF Z=3 GOTO 730
620 IF Z=4 GOTO 830
630 LET A2=0
```

```
640 LET P1=3.1415927
650 FOR I=1 TO M
660   LET X1= ABS (60-X[I])
670   LET Y=(90-X1)*P1/180
680   LET A= SIN (Y)
690   LET A1[I]=S[I]*A
700   LET A2=A2+A1[I]
710 NEXT I
720 GOTO  920
730 LET A2=0
740 LET P1=3.1415927
750 FOR I=1 TO M
760   LET X1= ABS (240-X[I])
770   LET Y=(270-X1)*P1/180
780   LET A= SIN (Y)
790   LET A1[I]=S[I]*A
800   LET A2=A2+A1[I]
810 NEXT I
820 GOTO  920
830 LET A2=0
840 LET P1=3.1415927
850 FOR I=1 TO M
860   LET X1= ABS (270-X[I])
870   LET Y=(270-X1)*P1/180
880   LET A= SIN (Y)
890   LET A1[I]=S[I]*A
900   LET A2=A2+A1[I]
910 NEXT I
920 LET A2=A2/M
930 LET A3=A1[1]
940 LET A4=A1[M]
950 LET S1=4*P1*P1*A2*A2/(W*W)
960 LET L3=S1/(A3*A3)
970 LET L4=S1/(A4*A4)
980 LET L2=L3-L4
990 LET L= SQR (L2)
1000 LET K2=4*P1*P1/(W*W)
1010 LET K= SQR (K2)
1020 LET C=(K2-L4)/L2
1030 LET C1= ATN ( SQR (1-C*C)/C)
1040 LET H1=P1-C1/(L* SIN (C1))
1050 IF L2>.9*P1*P1/(4*H1*H1) GOTO  1150
1060 PRINT
1070 PRINT "CRITERIA FOR LEE WAVES HAS NOT BEEN MET. DO YOU WISH"
1080 PRINT "TO CONTINUE THE CALCULATION. IF YES TYPE 1, IF NO TYPE"
1090 PRINT "2. CONTINUE IF STRONG INVERSION EXISTS OR IF WAVE"
1100 PRINT "CLOUDS ALREADY EXIST."
1110 PRINT
1120 INPUT R
1130 IF R=1 GOTO  1150
1140 IF R=2 GOTO  1770
1150 IF Z=1 GOTO  1190
1160 IF Z=2 GOTO  1220
1170 IF Z=3 GOTO  1250
1180 IF Z=4 GOTO  1280
1190 LET H=~~914~~ 692
1200 LET B=3379
1210 GOTO 1300
1220 LET H=~~914~~ 692
1230 LET B=756
1240 GOTO 1300
1250 LET H=~~1104~~ 830
1260 LET B=3702
1270 GOTO 1300
1280 LET H=~~2803~~ 2559
1290 LET B=4576
```

```
1300 LET P=2*P1*H*B*A1[1]*L2* SIN (C1)* SIN (C1)
1310 LET Q= EXP (K*B)*(P1-C1+ TAN (C1))
1320 LET W0= ABS (-P/Q)
1330 IF W0<W1 GOTO  1440
1340 IF W0>=W1 IF W0<W2 GOTO  1500
1350 IF W0>=W2 IF W0<W3 GOTO  1580
1360 LET C2=C1*180/P1
1370 PRINT
1380 PRINT "CALCULATION INDICATES SEVERE TURBULENCE PRESENT."
1390 PRINT "RECOMMEND SIGMET BE ISSUED. MAX TURBULENCE WILL OCCUR"
1400 PRINT "BELOW INVERSION IF PRESENT AND IN THE FIRST FEW WAVES"
1410 PRINT "NEAREST CREST OF RIDGE. IF NO INVERSION PRESENT MAX"
1420 PRINT "TURBULENCE WILL OCCUR BELOW LEVEL OF WAVE CLOUD TOPS."
1430 GOTO 1660
1440 PRINT
1450 PRINT "CALCULATION INDICATES LIGHT TURBULENCE OR LESS PRE-"
1460 PRINT "SENT VICINITY WAVE CLOUDS. UNLESS OTHER DATA INDICATE"
1470 PRINT "TURBULENCE OF A HIGHER CATAGORY. RECOMMEND NO AIRMET"
1480 PRINT "BE ISSUED."
1490 GOTO 1660
1500 PRINT
1510 PRINT "CALCULATION INDICATES LIGHT TO MODERATE TURBULENCE"
1520 PRINT "PRESENT VICINITY WAVE CLOUDS. RECOMMEND AIRMET BE"
1530 PRINT "ISSUED. MAX TURBULENCE WILL OCCUR BELOW INVERSION IF"
1540 PRINT "PRESENT AND IN THE FIRST FEW WAVES CLOSEST TO THE"
1550 PRINT "CREST OF THE RIDGE. IF NO INVERSION PRESENT MAX TUR-"
1560 PRINT "BULENCE WILL OCCUR BELOW LEVEL OF WAVE CLOUD TOPS."
1570 GOTO 1660
1580 PRINT
1590 PRINT "CALCULATION INDICATES MODERATE TO SEVERE TURBULENCE"
1600 PRINT "PRESENT VICINITY WAVE CLOUDS. RECOMMEND AIRMET BE"
1610 PRINT "ISSUED UNLESS OTHER DATA INDICATES A SIGMET IS RE-"
1620 PRINT "QUIRED. MAX TURBULENCE WILL OCCUR BELOW INVERSION IF"
1630 PRINT "PRESENT AND IN THE FIRST FEW WAVES CLOSEST TO THE"
1640 PRINT "CREST OF THE RIDGE. IF NO INVERSION PRESENT MAX TUR-"
1650 PRINT "BULENCE WILL OCCUR BELOW LEVEL OF WAVE CLOUD TOPS."
1660 PRINT
1670 PRINT "ALL OUTPUT VALUES ARE IN M.K.S. UNITS."
1680 PRINT
1690 PRINT "THE ABSOLUTE VALUE OF VERTICAL WIND SPEED = "W0;" M/S"
1700 PRINT
1710 PRINT "AVERAGE NORMAL WIND SPEED = "A2;" M/S"
1720 PRINT
1730 PRINT "AVERAGE STATIC STABILITY = "S1
1740 PRINT
1750 PRINT "WAVELENGTH = "W;" M"
1760 PRINT
1770 PRINT "DO YOU WISH TO CONTINUE? IF YES TYPE 1 IF NO TYPE 2."
1780 PRINT
1790 INPUT G
1800 PRINT
1810 IF G=1 GOTO  320
1820 CHAIN ""
1830 STOP
1840 END
```

APPENDIX II

SAMPLE OUTPUT

```
#TURBULENCE

THIS PROGRAM GIVES CATAGORIES OF TURBULENCE WHICH MAY
BE EXPECTED IN THE VICINITY OF WAVES IN THE LEE OF
MOUNTAIN RANGES OR INDIVIDUAL MOUNTAIN PEAKS. IT IS
CURRENTLY VALID FOR THE KOOLAUS WHEN WINDS ARE FROM
THE NE OR SW QUADRANTS, MT KAALA WHEN WINDS ARE FROM
SW QUADRANT AND MT HALEAKALA WHEN WINDS ARE FROM THE
W QUADRANT. THE INPUTS TO THE PROGRAM ARE THE NUMBER
OF PIBAL WINDS BETWEEN THE CREST OF THE MOUNTAIN
AND THE MAX WIND ABOVE THE CREST INCLUSIVE, THE WIND
DIRECTION AND SPEED AT EACH LEVEL, AND THE WAVELENGTH
MEASURED AS THE NUMBER OF WAVES PER DISTANCE, IE USING
AT LEAST 5 WAVE CLOUDS MEASURE THE DISTANCE FROM FIRST
TO LAST WAVE AND INPUT THE NUMBER OF WAVES AND THE
DISTANCE MEASURED IN NAUTICAL MILES (NM). THE 3000FT
WIND IS THE STARTING POINT FOR THE MOUNTAINS ON OAHU.
THE 10000FT WIND OR WIND NEAREST TO THAT LEVEL ABOVE
THE SUMMIT OF HALEAKALA IS THE STARTING POINT ON MAUI.
THE LAST WIND INPUTTED SHOULD ALWAYS BE THE MAX WIND.
USE THE BEST WINDS AVAILABLE. YOU MAY USE UP TO 25
WIND VELOCITIES. USE KNOTS FOR ALL WIND SPEEDS.
WHAT IS THE NUMBER OF WIND VELOCITIES YOU WISH TO USE?

? 3
GIVE ME YOUR VELOCITIES AS DIRECTION, SPEED RETURN.

? 080,25? 075,35? 075,36
GIVE ME NUMBER OF WAVES, DISTANCE MEASURED RETURN

? 6,35.4

GIVE ME WIND DIRECTION. IF NE QUAD TYPE 1, IF SW QUAD
TYPE 2, IF WAVES APPEAR TO BE FORMING IN THE LEE OF MT
KAALA ONLY TYPE 3, IF W QUAD TYPE 4.

? 1
CRITERIA FOR LEE WAVES HAS NOT BEEN MET. DO YOU WISH
TO CONTINUE THE CALCULATION. IF YES TYPE 1, IF NO TYPE
2. CONTINUE IF STRONG INVERSION EXISTS OR IF WAVE
CLOUDS ALREADY EXIST.

? 1
CALCULATION INDICATES LIGHT TURBULENCE OR LESS PRE-
SENT VICINITY WAVE CLOUDS. UNLESS OTHER DATA INDICATE
TURBULENCE OF A HIGHER CATAGORY. RECOMMEND NO AIRMET
BE ISSUED.

ALL OUTPUT VALUES ARE IN M.K.S. UNITS.

THE ABSOLUTE VALUE OF VERTICAL WIND SPEED =  1.47146  M/S

AVERAGE NORMAL WIND SPEED =  15.7989  M/S

AVERAGE STATIC STABILITY =  8.24256E-05

WAVELENGTH =  10933.9  M

DO YOU WISH TO CONTINUE? IF YES TYPE 1 IF NO TYPE 2.
```